

A táblázatkezelés alapjai

A „számolótábla” alapötlete

A szövegszerkesztő program mellett a leggyakrabban használt irodai alkalmazás a táblázatkezelő. Míg azonban a szövegszerkesztés egyidős az írásbeliséggel, addig a táblázatkezelő programok megjelenése a számítógépekhez kötődik.

A „számolótábla” alapötlete az, hogy a táblázat a kiszámolható adatok helyett a számítás módját megadó képletet tartalmazza. Például ha egy táblázatban a megvásárolandó tej mennyisége és egységára szerepel, akkor a fizetendő összeget nem kézzel számítjuk ki és írjuk be a táblázatba, hanem ezt maga a számolótábla végzi el a megadott képlet (*mennyiség és egységár szorzata*) alapján.

Az első széles körben elterjedt táblázatkezelő program, a *VisiCalc*, amelyet két egyetemista, Dan Bricklin és Bob Frankston készített, 1979-ben jelent meg, utolsó változatát 1985-ben adták ki. A táblázatkezelő elvi felépítése azóta sem változott.

	A	B	C	D
1	Termék	Mennyi	Egységár	Fizetendő
2	Tej	2	3.6	7.2
3	Kenyer	1	3.6	3.6
4	Kifli	3	.4	1.2
5	Sportsz	2	1.2	2.4
6	Fozokol	.2	28	5.6
7	Összes			20

	A	B	C	D
1	Termék	Mennyiség	Egységár	Fizetendő
2	Tej	2	319 Ft	638 Ft
3	Kenyer	1	249 Ft	249 Ft
4	Kifli	3	29 Ft	87 Ft
5	Sportszelet	2	89 Ft	178 Ft
6	Főzőkolbász	0,2	1 290 Ft	258 Ft
7	Összesen			1 410 Ft

► Napi bevásárlás VisiCalccal (1979-ből) és Excellel (2020-ban). A képletek majdnem ugyanazok, például a D2-es cellában VisiCalc esetén még $+B2*C2$ volt (ma $=B2*C2$ -t írunk).

A táblázatkezelő programok az adatok tárolása és látványos megformázása mellett lehetővé teszik számítások automatikus elvégzését, diagramok készítését. A mai táblázatkezelő programok támogatják ezenkívül üzleti előrejelzések készítését, egyenletek közelítő megoldását, illetve tartalmaznak adatbázis-kezelő funkciókat is.

Az asztali programok mellett egyre elterjedtebbek az online elérhető számolótáblák (pl. *Google Táblázatok*) és a mobilappok. Tanulás közben ezért a hangsúlyt az általános alapelvekre és funkciókra kell helyezni, hogy könnyen tudjunk közöttük váltani.

Ebben a fejezetben főleg két asztali táblázatkezelő programot, a *Microsoft Excelt* és a *LibreOffice Calcot* mutatjuk be példaként. Ha egy funkciót a két programban eltérően kell használni, akkor a szövegben először a *Microsoft Excelre* vonatkozó megoldás szerepel, ezt követi zárójelben a *LibreOffice Calc*-beli megvalósítás.

1. példa: A színjátszófesztivál bevétele

Az iskolai színjátszófesztivál gálaelőadását az Irka Iskola a Városi Színház nagytermében tartja. Ebben a példában az előadás bevételeit összesítjük.

Írjuk be az adatokat az ábrának megfelelően! Az adatok bevitele közben ügyeljünk arra, hogy a szomszédos cellákba a tabulátorgomb vagy a kurzormozgató billentyűk használatával lépjünk! Egy cella tartalmát utólag is javíthatjuk, ha például előbb kettőt kattintunk rá az egér bal gombjával.

Egy **cellára** vagy **cellatartományra** az oszlop- és sorazonosítók „sakkszerű” megadásával hivatkozhatunk. Az ábrán például éppen a C5-ös cellába írjuk be a karzatra szóló jegyek árát, míg a jegyek eladott darabszámát a hely függvényében az A2:B5 tartomány tartalmazza.

The image shows two side-by-side screenshots of an Excel spreadsheet. The spreadsheet has columns A, B, C, and D, and rows 1 through 6. The data is as follows:

	A	B	C	D
1	Hely	Darab	Ár (Ft)	Összeg
2	Földszint	124	3200	
3	Erkély	68	4200	
4	Páholy	16	7200	
5	Karzat	42	22	
6	Összesen			

The left screenshot shows the data being entered. The right screenshot shows the formula `=B2*C2` being entered into cell D2, and a green box indicating that this formula is being copied down to cells D3, D4, and D5.

► Adat bevitele a C5-ös cellába

► A D2-es cella másolása

Az adatok beírása után meghatározzuk, hogy mennyi volt a bevétel helytípus szerint. Például a földszintre eladott jegyek után a bevételt a B2-es és a C2-es cellák tartalmának összeszorozásával kapjuk, így a D2-es cellába a következő **képletet** írjuk:

$$=B2*C2$$

A D2-es cella alatti D3:D5 tartományba hasonló képletet kell bevinnünk. Ezt megtehetjük egyenkénti beírással is, de célszerűbb a képletet lefelé másolni: egérrel „megfogjuk” a D2-es cella jobb alsó sarkán lévő kis négyzetet, majd lefelé húzzuk a D5-ös celláig. Ekkor a **cellákba az eredeti képlet a másolás irányának megfelelően módosítva kerül**, vagyis ezúttal a sorok azonosítója mindig eggyel nő.

Végül határozzuk meg a teljes bevételt a D6-os cellában! Erre megoldás lehet az

$$=D2+D3+D4+D5$$

képlet beszúrása, de látható, hogy ez sok cella esetén nem használható. A táblázatkezelők ezért az ilyen esetekre **függvények** használatát teszik lehetővé. Például a D2:D5 tartomány celláinak összegét a SZUM függvény segítségével határozhatjuk meg a következő képlettel:

$$=SZUM(D2:D5)$$

A SZUM függvény a zárójelben szereplő tartomány celláinak értékét adja össze. Ha a tartományban szöveget tartalmazó cellák is vannak, azokat nem veszi figyelembe.

A táblázatkezelés alapfogalmainak áttekintése

A táblázatkezelő programok a táblázat **oszlopait** A-val kezdve betűkkel, **sorait** egytől kezdve számokkal azonosítják. Egy adott **cellára** az oszlop- és sorazonosítójával hivatkozhatunk, például C5.

Egy téglalap alakú **cellatartományt** a bal felső, valamint a jobb alsó sarkában lévő cellák azonosítójával adunk meg, pl. A1:B5.

Adatainkat a táblázatkezelő program egy **munkafüzetben** tárolja, amely egy vagy több **munkalapból** áll. A munkalapok között az alkalmazásablak alsó részén válthatunk. Az adatok mentésekor a teljes munkafüzet kerül a fájlba.

A munkafüzetben mindig egy olyan cella van, amelybe adatokat tudunk bevinni a billentyűzetről, ezt **aktív cellának** nevezik.

	A	B	C	D
1	Hely	Darab	Ár (Ft)	Összeg
2	Földszint	124	3200	396800
3	1. emelet	68	4200	285600
4	2. emelet	16	7200	115200
5	3. emelet	42	2200	92400
6	Osszesen			890000
7				

Az aktív cella tartalma (szerkesztőléc, beviteli mező)

Az aktív cella vagy cellatartomány (névmező, névdoboz)

Teljes munkalap kijelölése

Aktív cella

► Néhány fontos elem a táblázatkezelők képernyőjén

Az oszlopok azonosítása Z után kétbetűsre vált, rendre AA, AB stb.

Egy vagy több teljes sor vagy teljes oszlop is alkothat tartományt, ennek megadása például B:B vagy B:G.

Egy tartomány kijelölése többnyire az egérrel történik, de megtehetjük azt a SHIFT gomb nyomva tartása mellett a kurzormozgató billentyűkkel is. Egy sort vagy egy oszlopot kijelölhetünk, ha a megfelelő azonosítóra kattintunk. A teljes munkalap kijelöléséhez pedig az oszlopazonosítók előtt és a sorazonosítók fölött lévő téglalapra kell kattintanunk.

A munkalap nem korlátlan méretű. Az üres munkalapot (vagy egy kitöltött tartományt) a CTRL + kurzormozgató nyilak lenyomásával járhatjuk körbe.

Új munkalapot a képernyő alsó részén lévő ⊕ ikonra (illetve a + ikonra) kattintva adhatunk a munkafüzethez. A munkalap nevét a jobb egérgombbal rákattintva, helyi menü **Átnevezés** (illetve a *Munkalap átnevezése*) menüponttal módosíthatjuk.


Ha a cella **szöveget** tartalmaz, akkor annak tartalmát a táblázatkezelő program **balra** zárja automatikusan, ha pedig **számot**, akkor **jobbra**. Ha tehát egy szám típusú adat balra kerül, akkor azt valószínűleg hibásan írtuk be.

A képletek egyenlőségjellel kezdődnek. A **képlet eredményét a táblázatkezelő programok az adatok változása esetén automatikusan újraszámolják**. Az újraszámítást az F9 funkciógomb lenyomásával is kiválthatjuk.

2. példa: A táblázat formázása és diagram beillesztése

A táblázatot a szövegszerkesztésben már megismert módon formázhatjuk: a cellákat **szegélyezhetjük**, megváltoztathatjuk a **hátterszínét**, módosíthatjuk a **betű formátumát** és a cellák **igazítását**.

	A	B	C	D
1	Hely	Darab	Ár (Ft)	Összeg
2	Földszint	124	3200	396800
3	Erkély	68	4200	285600
4	Páholy	16	7200	115200
5	Karzat	42	2200	92400
6	Összesen			890000



▶ A táblázat megformázása betűformázás, igazítás, hátterszín, szegély alkalmazásával

▶ A diagram összetevőit módosíthatjuk: a cím átírása

A legtöbb ember számára áttekinthetőbbek az adatok, ha azokat diagramon ábrázoljuk. Szemléltessük ezért a bevételt a hely függvényében egy kördiagramon!

Ehhez az A1:A5 és C1:C5 tartományokat kell megadnunk. Jelöljük ki ehhez az A1:A5 tartományt, majd nyomjuk le a CTRL gombot, és annak nyomva tartása közben jelöljük ki a C1:C5 tartományt is!

A kijelölt tartományhoz a táblázatkezelő program automatikusan elkészíti a kördiagramot a *Beszűrés > Diagramok > Kör vagy peregdiagram* (illetve *Beszűrés > Diagram > Diagramtípusok > Torta*) pont választásával.

A kész diagramot az egérrel mozgathatjuk, a sarkok húzásával átméretezhetjük, illetve az egyes összetevőkre kattintva, azok megjelenését módosíthatjuk.

Cellák, sorok, oszlopok elrendezése

Szöveges adatoknál gyakori, hogy a szöveg nem fér el a cellában. Amíg a szomszédos cella üres, addig látszólag ott folytatódik, de ha oda is adat kerül, akkor a „kilógó” rész nem látszik. Ilyenkor *csökkentsük a betű méretét, vagy növelhetjük az oszlop szélességét* az oszlop betűjelét tartalmazó téglalap jobb szélének húzásával. Mivel ez a probléma igen gyakori, a táblázatkezelő programok lehetővé teszik azt is, hogy a szöveget automatikusan több sorra tördelve helyezzük el a cellában, például a *Sortöréssel több sorba* (illetve a *Szöveg tördelése*) ikonnal.

A címetek előnyös a táblázat fölött középre zárni. Legegyszerűbb, ha ilyenkor a megfelelő cellákat egyesítjük: ekkor a szöveg automatikusan középre igazodik. Ezt megtehetjük a megfelelő cellák kiválasztása után például a *Cellaegyesítés* (illetve a *Cellák egyesítése és középre zárása*) ikonnal.

Számok esetén nyilván nem járható út, hogy sokjegyű szám esetén annak egy része lemaradjon. Ezért ilyen esetben a táblázatkezelő program figyelmeztet: a cellába egy ##### karaktersorozat kerül.

Sorok és oszlopok beszúrását, törlését vagy méretének számszerű megadását a megfelelő menüpontokkal végezhetjük.

Ilyenkor értelemszerűen ki kell jelölnünk azt a sort vagy oszlopot, amelyet törölni akarunk, vagy amely elé szeretnénk új sort vagy oszlopot beszúrni.

Ezeket a műveleteket például a *Kezdőlap* > *Cellák* csoport ikonjaival (illetve a méretezést a *Formátum* menü, a beszúrást és törlést pedig a *Munkalap* menü pontjaival) végezhetjük el.

3. példa: Fagylaltgombócok ára

Készítsünk táblázatot, amely megkönnyíti a fagylaltárus feladatát: leolvassa meg tudja róla mondani, mennyit kell fizetnie a vevőnek a gombócok száma alapján!

A táblázat feliratait készítjük el a bal oldali ábrának megfelelően! A táblázatnak tartalmaznia kell a gombócok számát, ezt gyorsan megcsinálhatjuk kitöltéssel. Írjuk be a B5:B6 tartomány celláinak tartalmát, majd jelöljük ki, és húzzuk a B6-os cella jobb alsó sarkában lévő négyzetet lefelé!

	A	B	C
1			
2		Egységár: 250 Ft	
3			
4		Darab	Összeg
5		1	
6		2	
7			
8			
9			

▶ Kitöltés az egér húzásával

	A	B	C	D
1				
2		Egységár: 250 Ft		
3				
4		Darab	Összeg	
5		1	250 Ft =B5*C2	
6		2	0 Ft =B6*C3	
7		3	#ÉRTÉK! =B7*C4	
8		4	1 000 Ft =B8*C5	
9		5	0 Ft =B9*C6	

▶ Relatív hivatkozás az egységárra

	A	B	C	D
1				
2		Egységár: 250 Ft		
3				
4		Darab	Összeg	
5		1	250 Ft =B5*\$C\$2	
6		2	500 Ft =B6*\$C\$2	
7		3	750 Ft =B7*\$C\$2	
8		4	1 000 Ft =B8*\$C\$2	
9		5	1 250 Ft =B9*\$C\$2	

▶ Abszolút hivatkozás az egységárra

Mivel a fizetendő összeg az egységár és a gombócok számának szorzata, írjuk be a C5-ös cellába a megfelelő képletet: $=B5*C2$, majd másoljuk lefelé! Jóllehet a képlet egy gombócra helyes értéket ad, a többi esetben helytelen, vagy a program hibát jelez.

Nézzük meg, milyen képletek születtek másolásakor! A cellacímek a képlet másolásakor a másolás irányának megfelelően módosultak. A gombócok száma esetén ez így helyes is, de az egységár esetén nem, annak nem lett volna szabad változnia.

Az egységár cellacímét tehát másolás előtt rögzíteni kell, ezt a következő formában tehetjük meg: $=B5*\$C\2 . Másolás után ezúttal helyes eredményt kapunk.

A képletben lévő relatív cellahivatkozás a képlet másolásakor a másolás irányának megfelelően módosul, míg az abszolút cellahivatkozás nem változik.

Relatív cellahivatkozás esetén a táblázatkezelő nem a hivatkozott cellacímét tárolja, hanem annak az aktuális cellához viszonyított helyzetét. (Például: a gombócok száma egy cellával balra van). Abszolút cellahivatkozás esetén viszont a program a cella tényleges helyét tárolja.

Abszolút cellahivatkozás megadásakor a sor-, illetve oszlopkoordináta elé \$ jelet kell tennünk. Például az =B5*\$C\$2 képletben a B5 relatív, míg a \$B\$2 abszolút hivatkozás.

Szükség lehet táblázatok előállításához a vegyes cellahivatkozásra, amikor az egyik koordináta abszolút és a másik relatív. Például: \$B5 vagy C\$2.

Az ábrán a 10-es szorzótábla egy részletét látjuk. A táblázat vegyes cellahivatkozással készült.

A C3-as cellába a B3-as és C2-es cellák szorzata van. De vajon a cellahivatkozás melyik koordinátája abszolút és melyik relatív?

Mivel az egyik tényező a B oszlopban, a másik pedig a 2. sorban van, ezért ezt a két koordinátát kell rögzítenünk. Így tehát a C3-as cellába az =B\$3*C\$2 képlet került.

	A	B	C	D	E	F	G	H
1								
2			1	2	3	4	5	6
3		1	1	2	3	4	5	6
4		2	2	4	6	8	10	12
5		3	3	6	9	12	15	18
6		4	4	8	12	16	20	24

► Példa vegyes cellahivatkozásra

Feladatok

1. A bal oldali ábra az Irka Iskola 9. b osztályának éves pénzforgalmát tartalmazza. Az osztálypénz szeptember 1-jén 0 Ft-tal indult, utána havi bontásban látjuk a bevételek és kiadások összegét. Készítsük el a táblázatot! A hónapok oszlopát kitöltéssel vigyük be. Milyen képlet került a E4-es cellába?

	A	B	C	D	E
	Osztálypénz - havi bontásban				
1					
2			Bevétel	Kiadás	Egyenleg
3	Aug				0
4	Szept	80000	7800		72200
5	Okt	120000	13700		178500
6	Nov	40000	23780		194720
7	Dec	20000	13300		201420
8	Jan	10000	1800		209620

	A	B	C	D	E
1	Benzinár:	380	Fogyasztás:	7,4	l/100 km
2					
3	mikor	honnan	hová	út (km)	költség (Ft)
4	hétfő	Budapest	Kaposvár	190	5342,8
5	kedd	Kaposvár	Répcelak	179	5033,48
6	szerda	Répcelak	Tófej	102	2868,24
7	csütörtök	Lenti	Siófok	164	4611,68
8	péntek	Siófok	Tyukod	427	12007,24
9	Összesen:				29863,44

2. A jobb oldali ábrán egy utazó ügynök havi útnyilvántartását látjuk. Az ügynök a saját gépkocsiját használja, amely 100 km-en a D1-es cellában szereplő mennyiségű benzint fogyasztja. A benzin ára a B1-es cellában van. Milyen képlet kerül az E4-es és E9-es cellába? Az A4:A8 tartományban kitöltést alkalmazzunk!
3. A Fibonacci-sorozat első két tagja 1, majd minden következő tag az előző két tag összege. Képlet segítségével jelenítsük meg a sorozat első 20 tagját!

Számok, szövegek, logikai kifejezések kezelése

4. példa: A tehetségverseny értékelése (tizedesjegyek, ezres tagolás)

Az Irka Iskola diákönkormányzata minden évben tehetségversenyt hirdet, a legjobb osztály egy egynapos kirándulást kap jutalmul. A tantermeket három szempont szerint értékelik, minden szempontra legfeljebb 10 pont kapható.

Töltsük le a *terem* nevű fájlt a tankönyv weboldaláról, vagy gépeljük be az adatokat a szürke hátterű cellák kivételével!

	A	B	C	D	E	F	G	H	I	J	K
1	Osztály	9.nyek	9.a	9.b	10.a	10.b	11.a	11.b	12.a	12.b	Átlag
2	Tisztaság	5	6	10	9	9	9	4	6	7	7,22222
3	Dekoráció	9	8	10	3	10	7	7	9	9	8
4	Növények	8	2	9	7	7	5	8	8	8	6,88889
5	Összesen	22	16	29	19	26	21	19	23	24	22,1111
6											
7	Legjobb eredmény:			29		Második legjobb:		26		Osztályok száma:	
8	Legrosszabb eredmény:			16		Utolsó előtti:		19			10

► Az iskolai tehetségverseny statisztikai adatai

A leggyakoribb statisztikai függvények, a **SZUM()**, **ÁTLAG()**, **MAX()**, **MIN()** megadják a zárójelben lévő tartományban szereplő számok összegét, átlagát, legnagyobb értékét, illetve legkisebb értékét.



A **NAGY(tartomány; n)** és a **KICSI(tartomány; n)** függvények kétparaméteresek, megadják a tartomány n-edik legnagyobb, illetve legkisebb értékét.

Tehát a K2-es cellába az =**ÁTLAG**(B2:J2), a B5-ös cellába az =**SZUM**(B2:B4), a D7-es cellába az =**MAX**(B5:J5), a D8-as cellába pedig az =**MIN**(B5:J5) képlet kerül.

Hasonlóképp a H7-es cella tartalma az =**NAGY**(B5:J5; 2) képlet, a H8-as celláé pedig az =**KICSI**(B5:J5; 2) képlet. A **MAX** és **MIN** helyett is alkalmazhattuk volna a **NAGY** és **KICSI** függvényeket, például az =**NAGY**(B5:J5; 1) képlettel.

Az osztályok számát többféle módon is megadhatjuk. Például a B5:J5 tartomány számokat tartalmazó celláinak megszámlálásával: =**DARAB**(B5:J5), vagy a B2:J2 tartományban lévő nem üres cellák számának megadásával: =**DARAB2**(B2:J2).

A K2:K5 tartományban az átlagok különböző számú tizedesjeggyel szerepelnek. Jobban összehasonlíthatók az adatok, ha ezeket azonos számú, tipikusan két tizedesjegy pontossággal jelenítjük meg.

A tizedesjegyek számát például a *Tizedeshelyek növelése, csökkentése*  (illetve a *Tizedesjegy hozzáadása, törlése* ) ikonokkal változtathatjuk meg.

Fontos tudnunk, hogy a táblázatkezelő program továbbra is a pontos értékeket tárolja, és csupán megjelenítéskor formáz adott tizedesjegyre.

Legyen például két cella tartalma 3,33 és 2,22, ezek összege 5,55. Ha azonban mindhárom adatot tizedesjegyek nélkül jelenítjük meg, akkor látszólag a 2 és 3 összege 6 lesz. Ha az értékeket ténylegesen kerekíteni szeretnénk, akkor a **KERÉKÍTÉS** függvényt kell használnunk: például az A2-es cellában: =KERÉKÍTÉS(A1; 0).

A **KERÉKÍTÉS** függvénynek két paramétere van: az első a kerekítendő szám, a második pedig a tizedesjegyek előírt száma. Például 12,357 kerekítése 1 tizedesjegyre: **KERÉKÍTÉS(12,357; 1) = 30,4**. A tizedesjegyek száma lehet negatív is, ilyenkor a függvény a tizedesjeltől balra kerekít, például **KERÉKÍTÉS(12,345; -1) = 10**.

	A	B	C	D
1	2,22	3,33	5,55	C1: =A1+B1
2	2	3	6	C2: =A2+B2
3	A2-ben is 2,22 van			

	A	B	C	D
1	2,22	3,33	5,55	C1: =A1+B1
2	2	3	5	C2: =A2+B2
3	A2: =KERÉKÍTÉS(A1;0)			

► Az adatok látszólagos kerekítése formázással és pontos kerekítése függvény alkalmazásával

Nagyobb számok beírásánál gyakran használt formátum, hogy a számokat ezres tagolással és 2 tizedesjegy pontossággal jelenítjük meg, például: 12 345,67. Ezt általában egy ikonra való kattintással is elérhetjük, például a *Kezdőlap > Szám > Ezres csoport* (illetve *Eszköztár > Számformátum: Szám*).

Fontos tudnunk, hogy amíg Magyarországon az ezres elválasztójel a szóköz, a tizedesjel pedig a vessző, addig angol nyelvterületen az ezres elválasztójel a vessző, a tizedesjel pedig a pont.

5. példa: Az átlagkereset változása (százalék és pénznem)

Az alábbi ábrán a magyarországi átlagkereset növekedését látjuk 2000 és 2005 között. Töltjük le a *kereset* nevű fájlt a tankönyv weboldaláról, vagy gépeljük be az 1. és 2. sor, valamint az A oszlop adatait!

Határozzuk meg a 3. sorban, hogy az előző évihez képest mennyivel nőtt az átlagkereset; a 4. sorban azt, hogy az előző évihez képest hány százalékkal nőtt az átlagkereset, végül az 5. sorban azt, hogy 2000-hez képest hány százalékkal nőtt az átlagkereset! Végül formázzuk meg a táblázatot a mintának megfelelően!

	A	B	C	D	E	F	G
1	év	2000	2001	2002	2003	2004	2005
2	átlagbér	87 645 Ft	103 553 Ft	122 482 Ft	137 193 Ft	145 520 Ft	158 343 Ft
3	növekedés		15 908 Ft	18 929 Ft	14 711 Ft	8 327 Ft	12 823 Ft
4	növekedés aránya		18%	18%	12%	6%	9%
5	... 2000-hez képest		18%	40%	57%	66%	81%

► A havi bruttó átlagkereset változása Magyarországon 2000 és 2005 között a Központi Statisztikai Hivatal adatai alapján

A 2. sorban az adatok pénznem formátumban szerepelnek. Ezt elérhetjük úgy, hogy az adatok bevitelkor beírjuk a forint jelét is, de választhatjuk a *Kezdőlap > Szám > Könyvelési számformátum* (illetve *eszköztár > Számformátum: pénznem*) ikont is. A megjelenő tizedesjegyek számát csökkentjük nullára!

Az átlagkereset változásának megadása az előző évihez képest a C3-as cellában: $=C2-B2$; a változás aránya az előző évihez képest a C4-es cellában: $=(C2-B2)/B2$. Ha az 5. sorban másolható képletet szeretnénk alkalmazni, akkor a 2000-es adatokat rögzíteni kell, így a C5-ös cellába a következő képlet kerül: $=(C2-BS$2)/BS2 .

A 4. és 5. sorban az adatok százalék formátumban vannak megadva. Az adatok kijelölése után ezt például a *Kezdőlap > Szám > Százalék %* (illetve *eszköztár > Számformátum: Százalék %*) ikonnal, majd a tizedesjegyek számának növelésével állíthatjuk be.

Számformátumok

A táblázatkezelő programok a **tárolt számokat többféle formátumban meg tudják jeleníteni** (például pénznem, dátum formában), de ettől a tárolásuk változatlan.


A megjelenő formátumok gyakran országonként is eltérő szabványt követnek (például pénznem jele, tizedesjel, dátumforma), ezért a táblázatkezelők **az adott nyelvet-területen használt formátumokat az operációs rendszer beállításaiából veszik**.

Megjelenítéskor megadhatjuk a tizedesjegyek számát. Ilyenkor az adat adott tizedesjegyre formázva jelenik meg, de a program továbbra is a pontos értéket tárolja.

Az előző példában már megvizsgáltunk két számformátumot: a százalékot és a pénznemet. **A százalék formátum a szám százszorosát jeleníti meg**, így például 16-ból 1600% lesz.

A pénznemet gyakran megjeleníthetjük *könyvelői formátumban* is, ebben az esetben a tizedesjelek egymás alá kerülnek, függetlenül a tizedesjegyek számától.

Ugyan ikonok segítségével a leggyakoribb számformátumokat közvetlenül beállíthatjuk, de az összes lehetőség eléréséhez általában be kell lépniük a megfelelő párbeszédablakba, például a *Kezdőlap > Szám > Számformátum* (illetve a *Formátum > Cellák*) ponttal.



► Számformátumok beállítása (Microsoft Excel)

6. példa: Vendégszoba kiadása (dátum és idő)

Címer Gábor nyaranta rendszeresen kiadja siófoki nyaralóját, a bérlők a kereslettől függően eltérő összeget fizetnek az ott töltött éjszakák után. A takarítást egy helyi vállalkozó órabéren végzi, a fizetendő összeg szigorúan arányos a munkaidővel nem egész számú órák esetén is.

Töltsük le a tankönyv weblapjáról a *nyaralo* nevű fájlt, vagy vigyük be az adatokat az ábra alapján, a mintának megfelelően, de a szürke háttérű cellák kivételével!

Határozzuk meg először a vendégek után fizetendő összeget! Írjuk a C3-as cellába az =B3-A3 képletet! A cellában láthatóan az eltelt éjszakák száma jelenik meg, így adódik, hogy az E3-as cellába az =C3*D3 képlet kerüljön.

Ha ugyanezt elvégezzük a takarítónak fizetendő adatokkal, meglepő eredményre jutunk. Míg az I3-as cellában az =H3-G3 képlet helyes eredményt mutat, addig a J3-as cellába írt =I3*\$J\$1 képlet eredménye csupán 833 Ft lesz a várt érték helyett.

A megoldáshoz meg kell ismerkednünk a táblázatkezelők dátum- és időkezelésével. A hétköznapi életben használt rendszer ugyanis nemcsak bonyolult, de kultúrkörönként eltérő is; így egy új, logikus, de szokatlan megoldást alakítottak ki.

	A	B	C	D	E	F	G	H	I	J
1	Vendégek						Takarítás (óránként)			5 000 Ft
2	Érkezés	Távozás	Nap	Egy éj	Fizetendő		Kezd	Végez	Óra	Összeg
3	2020.06.19	2020.06.25	6	8 000 Ft	48 000 Ft		14:20	18:20	4:00	833 Ft
4	2020.06.26	2020.07.04	8	9 500 Ft	76 000 Ft		12:00	15:30	3:30	
5	2020.07.11	2020.07.14	3	12 000 Ft	36 000 Ft		14:00	18:15	4:15	
6	2020.07.15	2020.07.16	1	14 000 Ft	14 000 Ft		10:10	13:25	3:15	

	A	B	C	D	E	F	G	H	I	J
2	Érkezés	Távozás	Nap	Egy éj	Fizetendő		Kezd	Végez	Óra	Összeg
3	44001	44007	6	8000	48000		0,597	0,7639	0,167	833,33333

- ▶ Az időszámítás egysége a nap, így két időpont különbsége az eltelt napok száma. Az alsó ábrán a számformátumok eltávolítása után a ténylegesen tárolt értékeket látjuk.

A táblázatkezelő programok esetén az idő egysége a nap, az időszámítás kezdete pedig 1900. január 1. 0 óra 0 perc. Ez azt jelenti, hogy 1900. január 2-a 2-nek, 1900. február 1-je 32-nek felel meg.

Mivel a dátum egysége a nap, két dátum különbsége az eltelt napok számát adja.

Az adott napon belüli időt a szám tizedesjegyei tárolják. Ha a tárolt szám 32,5, akkor az ténylegesen 1900. február 1. 12:00-át jelenti.

Ha meg szeretnénk nézni, hogy a táblázatkezelő program ténylegesen milyen értékeket tárol, állítsuk át a táblázat számformátumát *Általánosra!*

Leolvasható, hogy a példában szereplő 2020. 06. 19. 44 001-nek, míg 2020. 06. 25. 44 007-nek felel meg, a két szám különbsége valóban 6.

A jobb oldali táblázatban láthatjuk, hogy az idő egysége is a nap. Mivel egy nap 24 órából áll, a J3-as cellába az =I3*\$J\$1*24 képletet kell írunk.

Végül érdemes megismerkednünk a **SZÖVEG** függvénnyel, amely a dátum egyes részeit szövegesen jeleníti meg. Két paramétere van: az első a dátum, a második a formátumkód. Például =SZÖVEG(2020.02.04;"nnnn") eredménye „kedd”. Az „n” kód a nap számát adja egy vagy két jegyre (4 vagy 14), az „nn” pontosan két jegyre (04), az „nnn” a rövidített nevet (K), végül az „nnnn” a teljes nevet (kedd). Hasonló módon kaphatjuk a hónap adatait is („h”, „hh” stb.).

További számformátumok, egyéni formátum kialakítása

A *Tudományos számformátum* a matematikában megismert normálalak megfelelője, csupán a kitevőt (exponens) egy E betűvel elválasztva a szám (mantissza) mellé írja. Például a $6,022 \cdot 10^{23}$ helyett 6,022E+23 jelenik meg.

A *Tört formátum* az adott tizedes törtet az általunk megadott pontossággal közönséges tört alakban adja meg. Mivel a pontosságot mi állítjuk be, ugyanaz a szám (0,1357) a beállításoktól függően lehet például 1/7, 8/59 vagy akár 65/479 is.

A felhasználó létrehozhat *Egyéni számformátumot* is. Ezt főleg mértékegységeket tartalmazó adatok megadásánál használjuk. Például ha egy adatot km/h egységben szeretnénk megjeleníteni, az alkalmazandó formátumkód: 0" km/h"

A formátumkódot a formázandó cellák kijelölése után például a *Kezdőlap* > *Szám* > *Számformátum* (illetve a *Formátum* > *Cellák*) ponttal elérhető párbeszédablakban az *Egyéni számformátumokhoz* írhatjuk be.

	A	B	C
1	Idő	Sebesség	Út
2	3 h	70 km/h	210 km
3	C2: =A2*B2		

- Egyéni formátum beállítása és megjelenése a B2-es cellában (LibreOffice Calc)

7. példa: Továbbjutás a színjátszó fesztiválon (*logikai típus*)

Az Irka Iskola színjátszó fesztiválja több fordulóban zajlik, alkalmanként három előadást mutatnak be. A zsűri az alábbi táblázatnak megfelelően négy szempont szerint értékeli, és dönt a továbbjutásról, a különdíjról, illetve hozzájárul a költségekhez is.

A továbbjutás feltétele, hogy az összpontszám 80-nál több legyen. Írjuk a G2-es cellába ennek feltételét: =F2>80. (Az egyenlőségjel a képletre utal.) A cellában megjelenik az így megadott logikai kifejezés értéke: IGAZ vagy HAMIS. A logikai értékeket a táblázatkezelők középre igazítják.

Az eredmény áttekinthetőbb, ha az IGAZ és HAMIS értékek helyett a HA függvény segítségével a „Továbbjut” szöveget, illetve az „” üres szöveget írjuk a cellába:

=HA(F2>80; "Továbbjut"; "")

	A	B	C	D	E	F	G	H	I	J
1	Előadás	Darabválasztás, szövegkönyv	Jelmezek, díszlet	Jelenetek	Színészi teljesítmény	Összesen	Továbbjut (logikai)	Továbbjut (Ha...)	Küöldíj (Logikai)	Költség-térítés
2	Macbeth	23	25	22	24	94	IGAZ	Továbbjut	IGAZ	10 000 Ft
3	Éretlenek	24	12	25	16	77	HAMIS		IGAZ	7 700 Ft
4	A kigyó	24	23	20	19	86	IGAZ	Továbbjut	HAMIS	10 000 Ft
5	G2: =F2>80 H2: =HA(F2>80;"Továbbjut";"") I2: =VAGY(B1=25; C2=25; D2=25; E2=25) J2: =HA(F2>80;10000;F2*100)									

- A továbbjutás, a különdíj és a költségterítés meghatározása logikai függvényekkel

A **HA** függvénynek három paramétere van: az első egy logikai kifejezés, a második a cella tartalma IGAZ, a harmadik a cella tartalma HAMIS érték esetén.

A táblázatkezelők használják az **ÉS**, illetve a **VAGY** logikai műveleteket. Ezeket függvényként kell beírni. Például ha különdíj jár arra az előadásra, amely valamely értékelési szempontra a maximális 25 pontot kapta, akkor egy lehetséges megoldás:

=VAGY(B2=25; C2=25; D2=25; E2=25)

A **J** oszlopban a költségtérítésre vonatkozó képlet alkalmazását látjuk, ezt a fentiek alapján elemezzük önállóan!

Szöveges adatok kezelése

A táblázatkezelők a szöveges adatokat alapértelmezetten a cellában balra zárják. Ha egy számot tehát a program balra zár, azt valószínűleg hibásan vittük be.

Gyakran van szükségünk arra, hogy a program egy számként is értelmezhető értéket mindenképpen szöveggként kezeljen. Például az 1.23 lehet egy termék azonosítója is, de ezt a legtöbb program automatikusan január 23-nak értelmezi. Ilyenkor egy aposztrófot kell az adat elé írni.

Ha egy szöveges adatot cellák értékéből és szövegekből kell összefűznünk, akkor az & műveleti jelet használjuk. Például: =C1&". helyezett: "&C2.

Feladatok

- Hány napot éltünk eddig? Életünk hány százalékában voltunk eddig jelenlegi iskolánk tanulói? Válaszoljunk ezekre a kérdésekre táblázatkezelő program segítségével! (Az aktuális dátumot az =MA() paraméter nélküli függvény adja meg.)
- Alkalmazzunk olyan képletet a 4. példában, amely a különdíjban részesülőket VAGY függvény használatával nélkül választja ki!
- Csaba bácsi 55. születésnapjára biciklit kapott, azóta naponta gyakorol. Heti útjait mobilapplikációval rögzítette. Készítsük el a táblázatot teljesítményének vizsgálatához! A távolságot km-ben, az időt óra:perc formában adjuk meg.
 - Határozzuk meg a **D** oszlopban a napi utak átlagsebességét!
 - Határozzuk meg a **C9**-es cellában a héten megtett teljes utat, a **C10**-esben pedig az egy napra jutó átlagos utat!
 - Írassuk képlettel a **D9**-es cellába a legnagyobb sebességet!
 - Az **E** oszlop cellába kerüljön „+” jel, ha aznap Csaba bácsi 20 km-nél többet tett meg, egyébként kerüljön „-” jel!
 - A **C** oszlopban alkalmazzunk *km*, a **D** oszlopban pedig *km/h* mértékegységet!
 - Az **F** oszlopban százalék formában jelenítsük meg, hogy az aznap megtett út hány százaléka volt a teljes hetinek!

	A	B	C	D	E
1	Indulás	Érkezés	Út	Sebesség	Több
2	14:21	15:35	14,25		
3	10:13	11:55	20,2		
4	9:49	11:22	19,6		
5	15:05	17:19	27,4		
6	14:15	16:13	25,55		
7	14:02	14:53	13,37		
8	13:59	15:57	21,91		

Diagramkészítés

Az alábbi táblázat a világ népességének alakulását szemlélteti az ENSZ adatainak alapján az elmúlt 50 évben, 5 éves periódusonként. Példánkban a táblázat adatait diagramok segítségével szemléltetjük. Az adatokat a *kontinensek* nevű fájlban találjuk.


	A	B	C	D	E	F	G	H	I	J	K	L
1		1970	1975	1980	1985	1990	1995	2000	2005	2010	2015	2020
2	Afrika	363	415	476	549	630	717	811	916	1 039	1 182	1 341
3	Ázsia	2 142	2 401	2 650	2 921	3 226	3 493	3 741	3 978	4 210	4 433	4 641
4	Európa	657	677	694	708	721	727	726	729	736	743	748
5	Észak-Amerika	231	242	254	266	280	294	312	327	343	357	369
6	Latin Amerika	287	323	361	402	443	483	522	558	591	624	654
7	Óceánia	20	22	23	25	27	29	31	34	37	40	43
8	Összesen	3 700	4 079	4 458	4 871	5 327	5 744	6 143	6 542	6 957	7 380	7 795

► A világ népességének változása az ENSZ adatai alapján

(Forrás: <https://population.un.org/wpp/Download/Standard/Population/> Utolsó letöltés: 2020.03.14.)


8. példa: Hogyan oszlott meg az egyes kontinensek között a Föld lakossága 2000-ben?

Az adatokat ezúttal a Föld teljes népességéhez viszonyítva szeretnénk ábrázolni, ezért készítsünk kördiagramot! A diagram elkészítéséhez nemcsak az adatokat, hanem a kontinensek nevét is ki kell jelölnünk, vagyis az A2:A7 és a H2:H7 tartomány celláit.


A diagram típusát a *Beszűrés > Diagramok > Kör- vagy peregdiagram beszurása*  menüponttal (illetve a *Beszűrés > Diagram > Diagramtündér* eszközzel) adhatjuk meg. Itt többféle lehetőségünk van: szokásos kördiagram (illetve tortadiagram), 3D kördiagram (illetve térhatású tortadiagram), körgyűrűcikkekkel álló peregdiagram (illetve fánkdiagram).

Ahhoz, hogy a diagram megfelelően elképzeléseinknek, további beállításokat is végezhetünk, ezeket programtól függően tehetjük meg. A *Microsoft Excel* menüje például kiegészül két új lehetőséggel: a *Diagramtervezéssel* és *Formátummal*, míg a *LibreOffice Calc*ban új menürendszert kapunk, ahol elsősorban a *Beszűrés* és a *Formátum* menüpontok lehetőségeire van szükségünk.

- Megadhatjuk, milyen elrendezésben jelenjenek meg a diagramon az egyes összetevők (cím, jelmagyarázat, adatok), például a *Diagramtervezés > Kész elrendezések* lehetőséggel (illetve a *Formátum* menü pontjaival).
- Megváltoztathatjuk a diagram egyes összetevőinek (címek, jelmagyarázatok) betűtípusát, szövegét, színét, illetve az egyes körcikkek színét, szegélyvonalát, ha a megfelelő elemre kattintunk.
- További összetevőket adhatunk hozzá, például a *Diagramtervezés > Diagram összetevő hozzáadása* (a diagramra kattintva > *Beszűrés* menü).
- Az összetevőket (pl.: diagramcím, adatfeliratok, egyes adatokat reprezentáló körcikkek) az egér húzásával mozgathatjuk.
- A diagramot más munkalapra helyezhetjük, például a *Diagramtervezés > Diagram áthelyezése* ikonnal (illetve a vágólapon át).
- 3D-s diagramok esetén pedig a diagramot térben elforgathatjuk, módosíthatjuk a perspektívát.



▶ Adatok viszonyítása egymáshoz és az összegükhöz képest kördiagram segítségével



▶ Két adatsor adatainak egymáshoz való viszonyítása oszlopdiagrammal

9. példa: Hogyan növekedett Európa és Afrika lakossága ebben az időszakban?

Ezúttal – feliratokkal együtt – az A1:L2 és A4:L4 tartományt kell kijelölnünk. Az adatokat oszlopdiagramon ábrázoljuk, mivel az lehetővé teszi két adatsor adatainak egymáshoz való viszonyítását.

Alapvetően két lehetőségünk van. *Csoportosított oszlopdiagram* (illetve *normál oszlopdiagram*) esetén az azonos évhez tartozó oszlopok egymás mellé kerülnek, így jól látható, hogy bár mindkét kontinens népessége nőtt, Afrikáé nagyobb ütemben. *Halmozott oszlopdiagram* esetén a két oszlop egymásra kerül.

Ezúttal is létrehozhatunk térhatású diagramot, illetve a két tengely felcserélésével úgynevezett sávdigramot.

Oszlopdiagramok esetén további beállítási lehetőségeink vannak.

- A tengelyekhez tengelycímetek adhatunk hozzá (Év, Millió fő), például a *Diagramtervezés > Diagram-összetevő hozzáadása > Tengelycímek* pont (illetve a diagram kiválasztása esetén a *Címek* gomb) segítségével.
- Megadhatjuk a skálabeosztást, vagyis a tengelyen mettől meddig, milyen osztásközzel jelenjenek meg az adatok. Ezúttal az adattartomány az y tengelyen 0-tól 1200-ig terjed, 100-as lépésközzel. A beállítást például a tengelyfeliratokra jobb gombbal kattintva a *helyi menü > Tengely formázása* ablakban érjük el. (Néha a diagramon az adatokat az adott tengelyen fordított sorrendben szeretnénk látni, ezt is itt állíthatjuk be.)
- Az oszlopdiagramok megjelenését esztétikusabbá teheti, ha az oszlopok alapértelmezett távolságát (átfedés) és távolságát a szélesség módosításával (térköz) beállítjuk. Az ábrán az átfedés 0, mivel az összetartozó oszlopértékek összeérnek. A beállítást elérhetjük például az oszlopokra kattintva a *helyi menü > Adatsorok formázása* menüpontjával.


10. példa: Európa lakosságának változása

Az ábrán Európa lakosságának időbeli változását grafikonon ábrázoltuk, ezt pontdiagram beszúrásával értük el. Ezúttal az A1:L1 és A4:L4 tartomány celláit jelöltük ki, és a *Pont görbített vonalakkal és jelölőkkel* (illetve a *Pontok és vonalak*, vonaltípus: *Sima*) lehetőséget választottuk.

A diagramon az oszlopdiaagramnál megismert változásokat állíthatjuk be a tengelyeken, de lehetőségünk van a vonal és a jelölők megválasztására is.

Érdekes lehetőség az adatokból matematikai úton számítható trendvonal felvétele, amely lehetővé teszi az értékek jövőbeni becslését. Ezt például az egér jobb gombjával a görbére kattintva a *Trendvonal felvétele* (illetve *Trendvonal beszúrása*) ponttal tehetjük meg. Beszúraskor kiválaszthatjuk a görbére illesztendő függvényt, ezúttal például a pontokhoz legközelebb álló egyenest kértük, és előírtuk a trendvonal egyenletének megjelenítését is.

Megjegyzés: Az adatokat ábrázolhatjuk vonaldiagramon is, de tudnunk kell, hogy a vonaldiagram az oszlopdiaagram logikáját követi, például az x tengelyen az értékek a sávok közepén jelennek meg.



► Grafikon trendvonallal és a trendvonal egyenletével

A diagramok áttekintése

Kördiagram: Csak egy adatsort mutat be, de lehetővé teszi az egyes adatok egymáshoz és az adatok összegéhez való viszonyítását is. Látványos változata a 3D kördiagram (illetve térhatású tortadiagram).




Oszlopdiaagram: Lehetővé teszi egy vagy több adatsor időbeli változásának követését és az adatok összehasonlítását is. Változatai a 3D (illetve térhatású) oszlopdiaagram, valamint a sávdiaagram.

Pontdiagram: Adatsorok változását a természettudományokban megszokott módon, grafikonon szemlélteti. Lehetőségünk van például trendvonalak felvételére is.

Feladatok

1. Ábrázoljuk grafikonon a világ népességének változását ebben az időszakban! Milyen trendvonal illeszkedik hozzá a legjobban? Állítsunk be a diagramterület háttérének egy képet a földgömbről!
2. Ábrázoljuk térhatású halmozott oszlopdiaagramon Észak-Amerika és Latin-Amerika lakosságának változását ebben az időszakban! Latin-Amerika adatai világos-, Észak-Amerika adatai pedig sötétvörös színben jelenjenek meg!

3. Ábrázoljuk 3D területdiagramon valamennyi kontinens lakosságának növekedését úgy, hogy az egyes kontinensekhez tartozó adatsorok ne takarják egymást!
4. Ábrázoljuk a *Számok, szövegek, logikai kifejezések kezelése* című fejezet első példájában szereplő adatokat (Teremszépségverseny) sugárdiagram (illetve hálódiaagram) segítségével!
5. Sajnos az adatok diagramokon történő szemléltetése sok esetben lehetővé teszi az adatok megtévesztő ábrázolását is. Az alábbi ábrákon erre látunk példákat.
 - a) Milyen eszközöket használtak az egyes diagramok készítői, és mit akartak „bizonyítani”?
 - b) Készítsük el *A világ népessége 2000-ben* című diagramot jól!



Problémamegoldás táblázatkezelővel

11. példa: Belépési adatok generálása

Egy cégnél a helyi hálózatba való belépéshez minden új dolgozó felhasználói nevet és egy egyszer használatos jelszót kap. A felhasználói név a vezeték- és utónév első két betűje, kiegészítve a születési év utolsó két jegyével, a jelszó pedig az utónév utolsó három karaktere, kiegészítve egy háromjegyű véletlen számmal.

Töltsük le a tankönyv weboldaláról az *account* nevű fájlt, vagy gépeljük be a táblázat első sorát és első két oszlopát!

	A	B	C	D	E	F	G	H	I
1	Név	Szülnap	Vez	Szk	Utó	Év	Felh	Vél	Jelszó
2	Füle Imre	1988.10.28	Fü	5	Im	1988	FüIm88	832	mre832
3	Megkő Tóni	1998.04.28	Me	6	Tó	1998	MeTó98	752	óni752
4	Gaz Ella	1992.05.07	Ga	4	El	1992	GaEl92	331	lla331
5	Ügyet Lenke	2001.01.01	Üg	6	Le	2001	ÜgLe1	370	nke370

► Belépési adatok generálása

Mintapéldánk megoldásához szükségünk lesz az alábbi szöveg-, illetve dátumkezelő függvényekre:

BAL(szöveg; darabszám), **JOB**(szöveg; darabszám): A szöveg bal, illetve jobb oldaláról adott darabszámú karaktert ad vissza.

KÖZÉP(szöveg; sorszám; darabszám): Visszaad adott darabszámú karaktert a szöveg adott sorszámú karakterétől kezdve.

SZÖVEG.KERES(mit; hol; honnan): Megkeresi a mit szöveg első előfordulását a hol szövegben a honnan sorszámú karaktertől indulva. Az utolsó paraméter elhagyható.

ÉV(dátum), **HÓNAP**(dátum), **NAP**(dátum): Megadják, hogy az adott dátum melyik év hányadik hónapjának hányadik napjára esik. Visszafelé: az év, hónap, nap számokból a dátumot a **DÁTUM**(év; hónap; nap) függvény határozza meg.

VÉL(): Egyenletes eloszlású véletlenszerű számot ad vissza, amely 1-nél kisebb, de 0-nál nem kisebb. Nincs paramétere.

VÉLETLEN.KÖZÖTT(alsó; felső): Az alsó és a felső érték közé eső véletlenszerű egész számot ad vissza.

A fenti függvények felhasználásával a vezetéknev első két betűjét a C2-es cellában az =BAL(A2; 2) képlet adja.

Az utónévhez meg kell keresnünk a szóköz helyét a D2-es cellában az =SZÖVEG.KERES(" "; A2) képlettel, ezt követi az utónév két karaktere az E2-es cellában: =KÖZÉP(A2; D2+1; 2).

A születési évet az F2-es cellában az =ÉV(B2) képlet adja, ennek utolsó két jegyét kell hozzáfűznünk az előző karakterekhez a felhasználói név előállításához:

=C2&E2&MARADÉK(F2; 100).

A jelszóhoz a H2-es cellában egy háromjegyű véletlenszámot generálunk a =VÉLETLEN.KÖZÖTT(100; 999) képlettel, amihez hozzá kell fűznünk a név utolsó 3 karakterét, így a jelszó az I2-es cellában az =JOBBA2;3)&H2 képlettel áll elő.

Érdeemes megemlíteni, hogy a véletlenszám-generátorok egy matematikai képlettel állítják elő az egyes tagokat, vagyis ezek valójában nem véletlenszerűek; ugyanakkor eloszlásuk hasonló a ténylegesen elvégzett kísérletekhez.

12. példa: Dolgozat értékelése

A 9. c osztály matematikadolgozatot írt, az eredményeket az A:B oszlopban látjuk. A dolgozatokat az elért pontszámok alapján a matematikatanár a G4:H8 tartományban szereplő segéd tábla felhasználásával számítja át osztályzatokra, például, aki 50 pontot elért, de 67 pontot már nem, az közepes (3) osztályzatot kap.

Töltsük le a *dolgozat* nevű fájlt a tankönyv weblapjáról, vagy gépeljük be az első sort, az első 2 oszlopot és a segéd táblát!

	A	B	C	D	E	F	G	H
1	Név	Pontszám	HOL.VAN	INDEX	FKERES		Dió Dina	67
2	Alma Ajna	84	4	jó	jó			
3	Barack Bardó	85	5	jeles	jeles			
4	Citrom Ciceró	66	3	közepes	közepes		0	elégtelen
5	Cseresz Nyeste	67	4	jó	jó		33	elégséges
6	Dió Dina	67	4	jó	jó		50	közepes
7	Eper Erik	49	2	elégséges	elégséges		67	jó
8	Füge Fürtike	50	3	közepes	közepes		85	jeles

► Példa keresőfüggvények használatára

A C oszlopban az osztályzatok számszerű értékét látjuk, vagyis azt, hogy a B oszlopban szereplő pontszám hányadik sávba esik a G4:G8 tartománnyal megadott sávok közül. Ezt a HOL.VAN függvénnyel érhetjük el. Példánkban a C2-es cellában a =HOL.VAN(B2; \$G\$4:\$G\$8) másolható képlet szerepel. Ez a függvény végighaladva a G4:G8 tartományon megáll a 4. elemnél (67), mert ez a B2-es cella értékénél (84) még kisebb, de a következő érték már nagyobb; és visszaadja a sáv sorszámát (4).

A D oszlopban az osztályzatok neveit látjuk. Ezeket a H4:H8 tartomány elemei adják meg; az 1. elem az elégtelen, a 2. az elégséges stb. Ezúttal az INDEX függvényt használjuk, amely visszaadja egy tartományból az adott sorszámú elemet. Példánkban a D2-es cella az =INDEX(\$H\$4:\$H\$8; C2) másolható képletet tartalmazza. Ez a „jó” szöveget adja vissza, mert C2 értéke 4, és ez a H4:H8 tartomány negyedik eleme.

A két képletet akár egymásba is ágyazhatjuk, így nem szükséges egy külön cellát igénybe vennünk. Ebben az esetben a D2-es cella tartalma:

=INDEX(\$H\$4:\$H\$8; HOL.VAN(B2; \$G\$4:\$G\$8))

A keresőfüggvények egy gyakori alkalmazását látjuk a G1:H1 tartományban. Ha a G1-es cellába beírjuk egy tanuló nevét, akkor a H1-es cellában az adott tanuló pontszáma jelenik meg.

A feladat ezúttal is megoldható az *INDEX...HOL.VAN* függvéypárossal. Az *A2:A8* tartomány celláiban a *HOL.VAN* függvénnyel megkeressük a kiválasztott tanuló helyét: *HOL.VAN(G1; A2:A8; 0)*, majd az „ennyiedik” adatot kiválasztjuk a *B2:B8* tartományból az *INDEX* függvény segítségével:

```
=INDEX(B2:B8; HOL.VAN(G1; A2:A8; 0))
```

A *HOL.VAN* függvénynek most van egy harmadik paramétere is, a 0. Ezzel érjük el azt, hogy pontosan a *G1*-es cellában lévő adatot keresse a függvény a megadott tartományban. Most – a sávokban való kereséssel ellentétben – nem kell rendezettnek lennie a megfelelő tartománynak.

Az eddigi feladatok megoldhatók más függvényekkel is, ezek közül a leggyakrabban használt az *FKERES*. Például az *E2*-es cellába a következő képletet is írhatjuk:

```
=FKERES(B2; $G$4:$H$8; 2)
```

Ez a következőt jelenti: keresse meg a program az *B2*-es cella értékét a *G4:H8* tartomány első oszlopában, és adja vissza a segédtábla 2. oszlopában lévő értéket. Az *FKERES* függvényt használhatjuk a *H1*-es cellában is, ebben az esetben is egy 0 paraméterrel kell jelezni a pontos keresést: *=FKERES(G1; A2:A8; 2; 0)*

Az *INDEX(tartomány; sor; oszlop)* függvény a *tartomány* megadott sorszámú sorából visszaadja az adott oszlopban lévő értéket. Ha az *oszlop* értéke 1, akkor a második paraméter elhagyható.

A *HOL.VAN(érték; tartomány; egyezés)* függvény megkeresi az *értéket* a *tartományban*, és visszaadja, hogy az hányadik. Ha pontos egyezést írunk elő, akkor az *egyezés* értéke 0. Ha azt a sávot keressük, amelyikbe az adott érték tartozik, akkor ez a paraméter elhagyható, de ekkor a tartománynak növekvően rendezettnek kell lennie.

Az *FKERES(cella; tartomány; k; egyezés)* függvény megkeresi a *cella* értékét a *tartomány* első oszlopában, és kapott sorból visszaadja a tőle jobbra, a *k.* oszlopban lévő értéket. Pontos keresésénél negyedik paraméterként 0-t kell megadnunk.

13. példa: Feltételhez kötött statisztikai számítások

Egy osztály életében rendszeresen visszatérő kérdés, hogy az angolosok vagy a németek teljesítenek jobban, gyakran készülnek olyan statisztikák, hogy mennyit hiányoznak az angolos fiúk, németes lányok stb. Az ilyen feladatokhoz a statisztikai függvények (*SZUM*, *ÁTLAG* stb.) egy olyan változatát kell használnunk, amelyeknél megadhatunk feltételeket is.

Töltsük le a tankönyv weboldaláról az *osztstat* nevű fájlt! A táblázatban a tanulók neve, neme, az általuk tanult első idegen nyelv, a félévi évfolyamdolgozat eredménye és az első félévben mulasztott órák száma szerepel az *A1:E26* tartományban.

Először határozzuk meg, hogy hány fiú és hány lány jár az osztályba! A feladatban a *B2:B26* tartományban kell megszámolnunk a „fiú”, illetve a „lány” szó előfordulásait. Ezt megtehetjük például a *DARABTELI* függvénnyel, amelynek két paramétere van: az első a vizsgált tartományt, a második a feltételt adja meg. Például a fiúk számát az

=DARABTELI(B2:B26;"fiú") képlettel határozhatjuk meg a H3-as cellában. A feladat megoldható másolható képlet alkalmazásával is: =DARABTELI(\$B\$2:\$B\$26;G3)

A H7-es cellában arra vagyunk kíváncsiak, hogy hány tanuló nem mulasztott egyetlen órát sem. Ehhez meg kell számolnunk az E2:E26 tartományban az üres cellák számát, amit a DARABÜRES függvényrel tehetünk meg: =DARABÜRES(E2:E26).

A DARABTELI függvényben egyenlőtlenséget is vizsgálhatunk, ilyenkor a relációs jelet idézőjelek közé kell tenni, és az & jellel kell hozzáfűzni a hivatkozott értéket, cellát vagy függvényt. Például azok száma, akik 10-nél több órát hiányoztak a H8-as cellában: =DARABTELI(E2:E26;">"&10), illetve az átlagosnál többet hiányzók száma a H9-es cellában: =DARABTELI(E2:E26;">"&ÁTLAG(E2:E26))

	A	B	C	D	E	F	G	H
1	Név	Nem	Nyelv	Jegy	Mulasztás			
2	Alma Ajna	lány	angol	4	2		A tanulók száma:	
3	Barack Bardó	fiú	angol	3	4		fiú	12
4	Citrom Ciceró	fiú	német	5			lány	13
5	Cseresz Nyeste	lány	német	4	14			
6	Dió Dina	lány	angol	4	5		Mulasztások száma:	
7	Eper Erik	fiú	angol	4	48		nem hiányzott	6
8	Füge Fürtike	lány	német	5	2		10-nél több óra	6
9	Galagonya Gala	lány	angol	4	3		átlagnál több óra	5
10	Grép Gerda	lány	német	3	66			
11	Josta Jolán	lány	angol	2	3		Angolos fiúk adatai:	
12	Kívi Kitti	lány	angol	4			tanulók száma	5
13	Körte Kötöny	fiú	német	5			együttes mulasztás	58
14	Licsi Liem	fiú	német	4	1		osztályzatok átlaga	3,80

► Feltételes számítások (a teljes osztály adatai az A1:E26 tartományban vannak)

Ha több feltétel együttes teljesülését vizsgáljuk, akkor a DARABTELI helyett a DARABHATÖBB függvényt kell használnunk. Ennek páros számú paramétere van: az első tartományt követi az első feltétel, majd a második tartományt a hozzá tartozó feltétel, és így tovább. A H12-es cellában az angolos fiúk számát keressük, vagyis azt, hogy a C2:C26 tartományban hányszor szerepel az „angol” szó, miközben a B2:B26 tartományban a „fiú” szó van: =DARABHATÖBB(C2:C26;"angol";B2:B26;"fiú").

Hasonlóan használjuk a SZUMHATÖBB és az ÁTLAGHATÖBB függvényeket, amelyek az első paraméterben megadott tartomány adataival végzik a megfelelő statisztikai számításokat. A feltételeket az ezt követő paraméterek tartalmazzák a DARABHATÖBB függvény-nél megadott módon. Például a H13-as cellában az =SZUMHATÖBB(E2:E26; C2:C26;"angol"; B2:B26;"fiú") képlet adja meg az angolos fiúk hiányzásainak összegét, míg ugyanezen tanu-lók évfolyamdolgozatának átlagát az =ÁTLAGHATÖBB(D2:D26; C2:C26; "angol"; B2:B26; "fiú") képlettel kapjuk. E két függvényhez hasonlóan használhatjuk a néhány programban már elérhető MAXHA és a MINHA függvényeket is.

Ha csupán egyetlen feltételt írunk elő, akkor használhatjuk a **SZUMHA** és az **ÁTLAGHA** függvényeket is, de esetükben ügyelni kell arra, hogy az összegzendő, illetve átlagolandó tartományt a harmadik paraméter, míg a kritériumot az első két paraméterük adja meg.

A **DARABHATÖBB**(*tart1; felt1; tart2; felt2; ...*) megszámlolja, hány olyan sor van egy adattáblában, amelyekre a *tart1* tartományában a *felt1*, a *tart2* tartományában a *felt2* ... feltétel teljesül. Ha egyetlen feltételünk van, akkor használhatjuk helyette a **DARABTELI**, az üres cellák megszámlolására pedig a **DARABÜRES** függvényt is.

A **SZUMHATÖBB**(*tart; tart1; felt1; tart2; felt2; ...*) egy adattábla *tart* tartományában összeadja azokat az elemeket, amelyekre a *tart1* tartományában a *felt1*, a *tart2* tartományában a *felt2* ... feltétel teljesül. Hasonlóan használhatjuk az **ÁTLAGHATÖBB**, a **MAXHA** és a **MINHA** függvényeket.

14. példa: Adatok kiemelése feltételes formázással

Ha fel akarjuk hívni a figyelmet bizonyos adatokra, akkor érdemes azokat a táblázatban eltérő formázással kiemelni. A táblázatkezelők a megadott feltételeknek megfelelő adatokat az előírt formátummal automatikusan ki tudják emelni, az adatok módosulása esetén pedig automatikusan megváltoztatják. Ezt a funkciót **feltételes formázásnak** nevezzük. Az egyszerűbb feltételeket általában kiválaszthatjuk a menü segítségével is, de az összetettebb feltételeket képlettel kell megadni, így most mi is ezzel a lehetőséggel fogunk megismerkedni.


Emeljük ki az osztály adatait tartalmazó táblázatban a jelesek félkövér és dőlt betűstílussal!

Jelöljük ki a **D2:D26** tartományt, majd válasszuk a **Kezdőlap > Feltételes formázás > Szabályok kezelése** (illetve a **Formátum > Feltételes formázás > Kezelés**) pontot! A megjelenő ablakban kattintsunk az **Új szabály > A formázandó cellák kijelölése képlettel** pontra (illetve a **Hozzáadás** gombra kattintva a megjelenő ablakban gördítsük le a **Feltétel 1** alatti listát, és kattintsunk a **képlet** lehetőségre)! Itt adjuk meg a képletet a tartomány első sorára vonatkoztatva: **=D2=5**, majd a **Formátum** gombra kattintva (illetve az **Alkalmazandó stílus** melletti lista legördítésével) állítsuk be a megfelelő formátumot! A táblázatkezelő program a megadott képlettel a teljes tartományt kitölti, vagyis a 3. sorban már az **=D3=5** feltételt vizsgálja, és így tovább.

Megjegyzések: Kicsit zavaró lehet, hogy a képletben két egyenlőségjel van, de az első csak kötelező bevezető eleme egy képletnek, nincs más szerepe. A **LibreOffice Calc** használata esetén az alkalmazandó formátumot – ha az nem szerepel a listán – a **Formátum > Stílusok > Új stílus** menüponttal hozhatjuk létre úgy, hogy egy cellát előbb megformázunk az előírt módon, majd annak kijelölése után ebben a pontban egy új nevet adunk a stílusnak.

Kicsit bonyolultabb a helyzet, ha az angolos tanulók sorait teljes egészében meg szeretnénk formázni, például egy halvány háttérszínnel. Ekkor előzetesen a teljes **A2:E26** adattáblát ki kell jelölnünk, míg az alkalmazandó képlet egy vegyes cellahivatkozást tartalmaz: **= $\$C2$ ="angol"**. Mivel a keresett szó a **C** oszlopban van, ezért a **C** oszlopazonosítót rögzítenünk kell, a sorazonosítót azonban nem szabad, hiszen a képletnek a tartomány 2. sora alatti sorokba másolva is működnie kell.

A formátumot a táblázatkezelők a kijelölt tartományra állítják be, de a feltétel vonatkozhat azon kívüli adatokra is. Például kiemelhetjük dőlt betűvel azok nevét, akik 10 óránál többet hiányoztak, ha a kijelölt tartomány az A2:A26, míg a feltétel: =E2>10.



► A feltételes formázás képletének megadása (balra Microsoft Excel, jobbra LibreOffice Calc)

Feladatok

1. A 11. példában szereplő táblázat *J* oszlopában állítsuk elő a felhasználók monogramját! Egészítsük ki a monogramot egy véletlenszerű nagybetűvel! (A karakterek kódja 65 és 90 közé esik, egy adott kódú karaktert a táblázatba a *KARAKTER* függvénnyel állíthatunk elő.)
2. Melyik tanuló hiányzott a legtöbbet a 9. c osztályban az *osztstat* fájl adatai alapján? Írassuk a tanuló nevét a *G16*-os, mulasztott óráinak számát a *H16*-os cellába!
3. Az angolosok vagy a németesek évfolyamdolgozata sikerült-e jobban? Határozzuk meg az *osztstat* fájl adatainak felhasználásával mindkét átlagot, majd egész mondatos választ írassunk ki a képernyőre! (Például: *A németesek értek el jobb eredményt.*)
4. Hogyan határozhatnánk meg a fiúk, illetve a lányok átlagos hiányzását csupán az *ÁTLAG* és a *HA* függvények felhasználásával?
5. Határozzuk meg képlet segítségével a németes lányok számát, együttes hiányzását és átlageredményét az *osztstat* fájl adatainak felhasználásával!
6. Emeljük ki piros színű betűkkel az *osztstat* táblázatban a lányok és kék színű betűkkel a fiúk sorait!
7. Neverstate-ben a rendőrség rendszeresen ellenőrzi az autók sebességét. Egyik alkalommal az ábrán látható adatokat mérték.
 - a) Tudjuk, hogy lakott területen 50 km/h, országúton 80 km/h, autópályán 110 km/h a megengedett legnagyobb sebesség. Írassuk képlettel a *D* oszlopba a megengedett legnagyobb sebesség értékeit a *B* oszlopban lévő adatok alapján!
 - b) A büntetés mértéke a megengedett sebességet meghaladó minden megkezdett 30 km/h többlet esetén 100\$, például aki lakott területen 120 km/h-val megy, az 300\$-t fizet. Írassuk ki képlet segítségével ezeket az összegeket az *E* oszlopba! (Mindkét feladatban használhatunk segéd táblát is.)

	A	B	C
1	rendszám	terület	mért
2	AA 333 B	lakott	45 km/h
3	DE 131 AD	országút	180 km/h
4	JA 144 SO	országút	50 km/h
5	IT 995 KT	pálya	250 km/h
6	TA 201 TT	országút	110 km/h
7	LE 333 LE	országút	88 km/h
8	KI 001 CH	lakott	72 km/h
9	BR 444 ER	országút	83 km/h
10	FC 010 SJ	országút	52 km/h

Fájlok kezelése, megosztása


Adatok importálása

A táblázatkezelő programok esetén az adatbevitel nem feltétlenül azok beírásával történik, hanem sok esetben az adatok már más formátumban rendelkezésre állnak.

Egy gyakori formátum a formázatlan szöveg (*txt*), amikor a cellák tartalmát egy adott soron belül tabulátorjelek választják el egymástól. Egy másik megoldás a *csv* (Comma-separated values, azaz vesszővel tagolt értékek) formátum, ahol egy sorban a cellák adatait vessző (a magyar változatban pontosvessző) választja el. Ezek a formátumok képleteket nem tartalmaznak. A legtöbb táblázatkezelő közvetlenül képes az ilyen fájlokat importálni. Ilyenkor ügyeljünk arra, hogy mentéskor a táblázatkezelő formátumában mentünk, különben a képletek és a formázási beállítások elvesznek!

Egy szövegszerkesztő programban készült táblázat adatait legegyszerűbben vágólapon keresztül tudjuk átemelni, de ezt a módszert használhatjuk a tabulátorokkal tagolt szöveg átvitelére egy egyszerű editorból, például a *Jegyzettömbből* is.

Az adatok gyakran az internetről származnak. A vágólapon ebben az esetben is működik, azonban az adatokat formázás nélkül érdemes beszúrni, például a *Kezdőlap > Beillesztés > Irányított beillesztés > Szöveg* (illetve a *Szerkesztés > Irányított beillesztés > Formázatlan szöveg*) lehetőséggel. Számok esetén problémát okozhat a tizedespont, illetve az ezres tagolásra használt nem törhető szóköz karakter. Ezeket a *Kezdőlap > Keresés és kijelölés > Csere* (illetve *Szerkesztés > Keresés és csere*) menüponttal cserélhetjük le, a tizedespontot tizedesvesszőre, a nem törhető szóközt pedig üres szövegre.



▶ Ugyanazokat a beállításokat érjük el, más módon (balra Microsoft Excel, jobbra: LibreOffice Calc)

Nyomtatás

A táblázat kinyomtatása általában nem egyszerű feladat, mivel ritkán fordul elő, hogy az adatok elrendezése igazodik a lap méretéhez. Másrészt táblázatkezelés közben nem szoktunk a papíralapú megjelenítés egyéb eszközeire: tájolás, margók, élőfej és élőláb tartalma, rácsvonalak, oszlop- és sorfejlécek megjelenítése stb. figyelni. Ezek beállítására a táblázatkezelő programok különböző lehetőségeket kínálnak, amelyeket például a *Fájl > Nyomtatás > Oldalbeállítás* (illetve a *Formátum > Oldal*) ponttal érhetünk el. Gyakran lehetőségünk van a táblázatkezelés logikáját követő normál elrendezés helyett közvetlenül a nyomtatási képet használni, például *Microsoft Excelben* a *Nézet > Lapelrendezés* menüpontra kattintva.

A legfontosabb beállítás a táblázat átméretezése úgy, hogy az kinyomtatva is áttekinthető legyen. Ehhez egyrészt megadhatjuk a táblázat átméretezésének arányát százalékban kifejezve, de gyakran azt is, hogy a táblázat kinyomtatva hány oldal széles és hány oldal magas legyen. Még ilyenkor sem biztos, hogy a laphatárok jó helyre kerülnek, ezért az egér húzásával egyenként módosíthatjuk az oldaltörések helyét, például a *Nézet > Oldaltörés előnézet* (illetve a *Nézet > Oldaltörés*) menüponttal.

Nagy méretű táblázatok esetén gyakori az is, hogy nem szeretnénk a teljes munkalapot kinyomtatni, hanem annak csupán egy részét. Ezt a nyomtatandó terület kijelölése után a *Lapelrendezés > Nyomtatási terület > Nyomtatási terület kijelölése* lehetőséggel állíthatjuk be (illetve közvetlenül nyomtatás előtt adhatjuk meg a *Fájl > Nyomtatás > Tartományok és példányszám > Kijelölt cellák* beállítással).

15. példa: Táblázat közös használata

Utolsó példánkban egy osztálybulit szervezünk. Osztályfőnöki órán már megállapodtunk ennek helyéről és időpontjáról, most azt szeretnénk egyeztetni, hogy ki jön el, és aki jön, mit hoz (üdítő, sós süti, édes süti).

Készítsünk táblázatot a mintának megfelelően, mentjük el egy felhőalapú tárhelyre, és osszuk meg a hivatkozást az osztályban! A megosztott táblázat a hivatkozás ismeretében szerkeszthető a böngészőből megnyitva webes alkalmazással vagy a mobiltelefonra telepített mobil alkalmazással is. A megosztott táblázatot egyszerre többben is szerkeszthetik, ilyenkor tiszteletben kell tartanunk a többiek munkáját, nehogy véletlenül töröljük vagy felülírjuk azt.

Sajnos asztali táblázatkezelő esetén a táblázat letöltés nélkül gyakran nem szerkeszthető, ilyenkor a módosított táblázatot vissza kell tölteni.

	A	B	C	D	E
		Jövök	Mit hozok?		
			Üdítő	Sós süti	Édes süti
3	Alma Ajna	nem			
4	Barack Bardó	x	x		x
5	Citrom Ciceró	x		x	
6	Cseresz Nyesta	x			
7	Dió Dina	x			x
8	Eper Erik	x			x
9	Füge Fürtike	x			
10	Galagonya Gala	nem			
11	Gráp Gerda	x		x	
12	Josta Jolán	x			
13	Kivi Kitti	x		x	
14	Körte Kotony	x	x		
15	Licsi Liem	nem			

▶ Megosztott táblázat szerkesztése (Android, Google Táblázat)

Feladatok

1. Budapest kerületei

- Keressük meg az interneten a budapesti kerületek adatait tartalmazó táblázatot, és válogáson át illesztjük be az általunk használt asztali táblázatkezelő programba formázási adatok nélkül!
- A táblázatot formázzuk meg, és nyomtassuk ki PDF-fájlba egyoldalásra!